

# ***The Gantt Chart on Steroids – Excel Project Manager Deluxe:***

## ***Free Microsoft Office Tutorials***



***Project tracking is a key requirement for the success of any good project.***

Excel Project Manager is an awesome application for anybody who is working project and who wants to track its progress over time. Break your project into manageable pieces and track their progress. This Microsoft Excel 2013 application is run completely with formulas and hyperlinks.

***We are creating a custom built Gantt chart that will track your project over 364 days (or whatever time you want).***



***This is a must see project.***

You will not believe how effective this can be. Project planning tools like this are hard to find. Create your own Project Planner.

There is so much to learn here. ***Do not miss this project.*** Learn how to adjust the application to suit your needs. I am more than happy to answer any questions that you may have if I can.

***Please use the contact form to do so. Don't forget to login to get the download template files below.***

***This project will work with Excel 2003 / 2007 / 2010.***

***Goal: Develop improved skills in Excel application development.***

***Features of the Project Planner Deluxe***

***Enhanced project planning***

***Assign and track project components over time***

***Check status at any moment in time***

***Report project overview***

***Identify weaknesses in the project development***

***any many more***

***What will we learn from this project?***

***Named ranges static and dynamic***

***Formula functions (IF,AND,OR,MIN,WEEKDAY,COUNTIF)***

***Conditional formatting formulas***

***Data validation***

***Linked images***

***Charts***

***And much more***

***What is included in the template?***

The key element to getting this project to work properly is the sheet formatting. I have supplied this in the template. I have also left the sheet navigation in for you. I have also left the entire cell formatting intact. The protection for the appropriate cells is set up for you. All that is need is to add the formulas and you're up and running.

***Watch this video to see the features of the Excel Project Manager in action.***

<http://www.youtube.com/watch?v=FOoWZalFUxU>

*This application has been designed by Trevor Easton for training purposes. You are able to use this for your personal use. The application as is or modified in not permitted for sale in any form. No warranties are implied or given with this application.*

***Online PC Learning is committed to providing free office tutorials***

***Our first piece of project work is related to the dates at the top of the Gantt chart***

***Watch this video to see the features of the Excel Project Manager in action.***

***Here are the steps to follow***

1. Add sequential number from P5 to NP5 starting at 1( you should end up with 364 sequential numbers)
2. ***Start date cell in I3***  
Add a start date to the project
3. ***Today's date in I2***  
Add today's date to the project by adding this formula ***=TODAY()***
4. ***End date in I4***  
Add an end date to the project ***=IF(I3="", "", NJ9+6)***
5. ***Total Days in cell K2***  
***=NP5 & " " & "Days"***
6. ***First day of the week in cell Q9***  
***=IF(WEEKDAY(I3,2)=1,I3,I3-WEEKDAY(I3,2)+1)*** to cell P9
7. ***Weeks Add this formula to X9***  
***=Q9+7*** and copy across to NP9
8. ***Month add this formula in cell Q7***  
***=Q9*** and scroll over to IU7 the custom formatting in the cell will only show the month part of the date
9. ***Year in cell Q6 add***  
***=Q9*** and X6 = X9 copy X6:X9 across to the end of the date block
10. ***Finish Date row 11. Add this to I11***  
***=IF(G11="", "", G11+H11)***
11. ***Cal Start in cell M11 add the formula below(this calculates the days from the start of the project)***  
***=IF(G11="", " ", G11-\$Q\$9+1)***
12. ***Active in cell J11 we need to see the status***  
***=IF(AND(G11>0,L11="",G11<\$I\$2),"In Progress",IF(L11>0,"Completed",IF(G11>0,"Not Started", "")))***
13. ***End Cal in cell N11 add the formula below(this calculates the days from the end of the project to the end of the sub element)***  
***=IF(I11="", "", I11-\$Q\$9+1)***
14. ***<> Today in cell O11(This formula will calculate the number of days from today to the start of the project )***  
***=IF(OR(G11=0,I11=0),0,\$I\$2-G11)***
15. ***Days add this to P11(Counts the number of days in this part of the project)***  
***=COUNTIF(Q11:NP11,">0")***
16. ***Finish Date row 10 add to cell I10 (returns the latest date in the range)***  
***=IF(I11="", " ", MAX(I11:I15))***
17. ***Status add to cell J10 (determines if this section has started)***  
***=IF(OR(\$I\$3="", \$I\$2>\$I10,),"Plan Expired",IF(\$G11>1,"Active", "Not Started"))***
18. ***% add this to cell K10 to give the percentage of time that has elapsed for the element***  
***=IF(P10=0, "", MIN(O10/P10,1))***
19. ***% add this to cell K11 to give the percentage of time that has elapsed for the sub element***  
***=IF(OR(P11=0,G11>\$I\$2), "", MIN(O11/P11,1))***
20. ***Cal Start in cell M10 add the formula below(this calculates the earliest date in the sub elements)***  
***=MIN(M11:M15)***
21. ***Cal End in cell N10 add the formula below(this calculates the latest date in the sub elements)***  
***=MAX(N11:N15)***

22. <> Today in cell O10(This formula will calculate the number of days from today to the start of the project )  
=IF(OR(G10=0,I10=0,G10>\$I\$2),0,\$I\$2-G10)
23. Days add this to P10(Counts the number of days in this part of the project)  
=COUNTIF(Q10:NP10,">0")
24. In Q10 add this formula( adds the number one to the date period  
=IF(AND(Q\$5>=\$M10,Q\$5<=\$N10),1,"") and scroll or copy this formula to NP10
25. In Q11 add this formula( adds the number 1/ 2 or 3 to the date period  
IF(AND(Q\$5>=\$M11,Q\$5<=\$N11),IF(\$J11="Completed",3,IF(\$J11="In Progress",2,1)),"") and scroll or copy this formula to IV11

## **The data validation and dynamic named ranges are added in the order below**

**Watch this video to see the features of the Excel Project Manager in action.**

<http://www.youtube.com/watch?v=1X6BAnL1lqs>

[As a refresher you may want to watch this video on dynamic named ranges](#)

1. **Dynamic list of Team Members on the Interface sheet**  
Add a List at S6 of team members and create a dynamic named range called "Team"  
=OFFSET(Interface!\$S\$6,,,COUNTA(Interface!\$S\$6:\$S\$13))
2. **Dynamic list for Progress on the Interface sheet**  
Add a List at T6 of progress items and create a dynamic named range called "Progress"  
=OFFSET(Interface!\$T\$6,,,COUNTA(Interface!\$T\$6:\$T\$13))
3. **Dynamic list for Duration on the Interface sheet**  
Add a List at U6 of duration in multiples of 7 use a formula = previous cell +28 to the value of 334 and create a dynamic named range called "Duration"  
=OFFSET(Interface!\$U\$6,,,COUNTA(Interface!\$U:\$U))
4. **Dynamic list of Elements on the Interface sheet**  
Add a List at P6 of team members and create a dynamic named range called "Elements"  
=OFFSET(Interface!\$P\$6,,,COUNTA(Interface!\$P\$6:\$P\$35))
5. **Dynamic list of Sub Elements on the Interface sheet**  
Add a List at Q6 of team members and create a dynamic named range called "SubElements"  
=OFFSET(Interface!\$Q\$6,,,COUNTA(Interface!\$Q\$6:\$Q\$35))
6. **On the Strategic Plan sheet**  
Add a data validation list in F11 for team members using the dynamic named range call "Team"
7. **On the Strategic Plan sheet**  
Add a data validation list in J11 for the progress items using the dynamic named range call "Progress"
8. **On the Strategic Plan sheet**  
Add data validation list in H11 for duration items and allow manual entry
9. **Add data validation in cells G11**  
as follows Data /Data validation/Settings tab/allow: Date/Data:Between/Start Date:=\$I\$3/End Date:=\$I\$4 then add an input and error message


## Conditional formatting is going to bring this project to life

Watch this video to see the features of the Excel Project Manager in action.

<http://www.youtube.com/watch?v=UO1TagtSOzY>


1. **Add conditional formatting to in J11**

1. Format cells only that contain /format only cells/ Cell value /equal to/ = “Completed” format the color to a green background cell interior
2. Format cells only that contain /format only cells/ Cell value /equal to/ = “In Progress” format the color to a dull orange background cell interior
3. Format cells only that contain /format only cells/ Cell value /equal to/ = “Not Started” format the color to a red background cell interior




2. **In the range C10to P10 at conditional formatting**

1. Format cells only that contain /use formula to determine which cells to format/ Format values where this formula is true: / =\$J10=“Plan Expired” format the color to a red background cell interior
2. Format cells only that contain /use formula to determine which cells to format/ Format values where this formula is true: / =\$J10=“Active” format the Green to a dull orange background cell interior



3. **Select cells Q10 to NP10 and add this conditional formatting**

1. Format cells only that contain /format only cells/ Cell value /equal to/ = 1 format the color to a dark blue background cell interior with a striped pattern style




4. **Select cells Q11 to NP11 and add this conditional formatting**

1. Format cells only that contain /format only cells/ Cell value /equal to/ = 1 format the color to a red background cell interior
2. Format cells only that contain /format only cells/ Cell value /equal to/ = 2 format the color to a dull orange background cell interior




3. Format cells only that contain /format only cells/ Cell value /equal to/ = 3 format the color to a green background cell interior



5. **Copy the completed rows**  
Highlight row 11 and copy to row 20
6. **Copy and paste special the comments**  
in in columns G / I / L to row 15
7. **Add the full range of data**  
Copy rows 10 to 15 and paste as many times as needed for the project
8. **Add navigation buttons**  
Add a shape for navigation to the left of the data
  1. Create a static named range in cell C10 called "Top"
  2. Right click the shape and assign a hyperlink to the named range "Top"
9. **Add quick scroll buttons**
  1. Create a static named range in cell Q10 called "Start"
  2. Create a static named range in cell NP10 called "Finish"
10. **Freeze panes**  
in cell Q6 freeze pane
11. **Select cells I11 to I15 and add this conditional formatting**  
=AND(\$I11>\$I\$4, \$G11>0)





## ***Linked picture and pie chart***

***We will now add the finishing touches to the project***

***Watch this video to see the features of the Excel Project Manager in action.***

<http://www.youtube.com/watch?v=siU-bYQLgOk>

## ***On the Interface sheet***



### ***On the Interface sheet in cell F8***

***=COUNTIF(Strategic\_Plan!\$J\$11:\$J\$33,Interface!F7)***

### ***On the Interface sheet in cell G8***

***=COUNTIF(Strategic\_Plan!\$J\$11:\$J\$33,Interface!G7)***

### ***On the Interface sheet in cell H8***

***=COUNTIF(Strategic\_Plan!\$J\$11:\$J\$33,Interface!H7)***

### ***On the Interface sheet in cell F8***

***=SUM(Strategic\_Plan!P11:P33)&" " &"Days"***

## ***Adding the linked picture***

1. On the strategic plan copy the area that you want to link.
2. Move to the interface sheet and select the paste icon drop down arrow in the ribbon and choose Picture Link
3. Resize and format to suit your taste.






Project Manager									
Strategic Plan									
	Activity	Start Date	End Date	Duration	Project Manager	Status	% Complete	Cost	Revenue
1	Project Manager	1/1/2013	31/12/2013	365	John Doe	Completed	100%	1000	1000
2	Project Manager	1/1/2013	31/12/2013	365	John Doe	Completed	100%	1000	1000
3	Project Manager	1/1/2013	31/12/2013	365	John Doe	Completed	100%	1000	1000
4	Project Manager	1/1/2013	31/12/2013	365	John Doe	Completed	100%	1000	1000
5	Project Manager	1/1/2013	31/12/2013	365	John Doe	Completed	100%	1000	1000
6	Project Manager	1/1/2013	31/12/2013	365	John Doe	Completed	100%	1000	1000
7	Project Manager	1/1/2013	31/12/2013	365	John Doe	Completed	100%	1000	1000
8	Project Manager	1/1/2013	31/12/2013	365	John Doe	Completed	100%	1000	1000
9	Project Manager	1/1/2013	31/12/2013	365	John Doe	Completed	100%	1000	1000
10	Project Manager	1/1/2013	31/12/2013	365	John Doe	Completed	100%	1000	1000

### Add the pie chart

1. Select F7 to H8 and choose the Insert tab and select the pie chart
2. Format the chart to suit your personal taste.



### The finishing touches

#### Form the file menu

Remove the sheet tabs

#### Select the View tab and

Hide the headers

Hide the formula bar

***On the review tab***

Protect the worksheets

There has been a lot to learn in the project management tracker. Every project is different and because of this it will be necessary to adjust this program to suit your personal or business situation. I feel confident that I have given you enough information to do this through the videos and information on the website.

Best wishes with your learning endeavors

Trevor